Kinematyka i Dynamika Robotów

Instrukcja do zajęć laboratoryjnych

dr inż. Dawid Pietrala
dpietrala@tu.kielce.pl

1 Laboratorium 3: Macierze przekształceń jednorodnych

1.1 Cele zajęć

Celem laboratorium jest zapoznanie studentów z macierzami przekształceń jednorodnych oraz ich zastosowaniem w opisie położenia i orientacji układów współrzędnych w przestrzeni dwuwymiarowej i trójwymiarowej.

Podczas zajęć studenci nauczą się:

- łączyć translację i rotację w jednej macierzy,
- interpretować geometrycznie przekształcenia pomiędzy układami współrzędnych,
- składać kolejne przekształcenia w łańcuch kinematyczny,
- wyznaczać położenie punktów w różnych układach odniesienia,
- opisywać pozycję i orientację układów w przestrzeni 2D i 3D,
- wykorzystywać macierze przekształceń jednorodnych w prostych zadaniach z zakresu kinematyki robotów.

1.2 Wprowadzenie teoretyczne

W robotyce bardzo często zachodzi potrzeba jednoczesnego opisu przesunięcia oraz obrotu. Sama macierz rotacji opisuje wyłącznie zmianę orientacji, natomiast sam wektor przesunięcia opisuje wyłącznie zmianę położenia. Z punktu widzenia analizy kinematyki robotów wygodniej jest korzystać z jednego obiektu matematycznego, który zawiera obie te informacje naraz. Takim obiektem jest macierz przekształcenia jednorodnego.

W niniejszej instrukcji przyjmujemy następującą konwencję zapisu:

$$^{i}H_j$$

Macierz $^{i}H_j$ opisuje położenie i orientację układu j względem układu i. Inaczej mówiąc:

- indeks górny i oznacza **układ odniesienia**,
- indeks dolny j oznacza **układ rozpatrywany**.

Taka konwencja jest bardzo wygodna, ponieważ od razu wiadomo, względem jakiego układu opisywany jest dany układ lokalny. Przykładowo:

$$^{0}H_1$$

oznacza macierz opisującą układ 1 względem układu bazowego 0.

Jeżeli chcemy przejść od układu 0 do układu 2 przez układ pośredni 1, to składanie przekształceń zapisujemy jako:

$$^{0}H_2 = {}^{0}H_1\,{}^{1}H_2$$

Taki zapis jest naturalny, ponieważ środkowy indeks się „zgadza”: pierwsza macierz kończy się na układzie 1, a druga zaczyna się od układu 1.

Macierz przekształcenia jednorodnego w 2D

W przestrzeni dwuwymiarowej macierz przekształcenia jednorodnego ma postać:

$$^{i}H_j = \begin{bmatrix} \cos\varphi & -\sin\varphi & x \\ \sin\varphi & \cos\varphi & y \\ 0 & 0 & 1 \end{bmatrix}$$

gdzie: x i y opisują przesunięcie początku układu j względem układu i, natomiast φ opisuje orientację układu j względem układu i.

Punkt na płaszczyźnie zapisujemy w postaci jednorodnej:

$$p = \begin{bmatrix} x \\ y \\ 1 \end{bmatrix}$$

Jeżeli punkt jest opisany we współrzędnych układu j, to jego współrzędne w układzie i wyznaczamy jako:

$${}^{i}p = {}^{i}H_j\,{}^{j}p$$

Macierz przekształcenia jednorodnego w 3D

W przestrzeni trójwymiarowej macierz przekształcenia jednorodnego ma postać:

$${}^{i}H_j = \begin{bmatrix} & & & x \\ & {}^{i}R_j & & y \\ & & & z \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

gdzie ${}^{i}R_j$ jest macierzą rotacji 3×3, opisującą orientację układu j względem układu i, natomiast x, y, z są współrzędnymi początku układu j względem układu i.

Punkt zapisujemy wtedy jako:

$$p = \begin{bmatrix} x \\ y \\ z \\ 1 \end{bmatrix}$$

a transformacja ma postać:

$${}^{i}p = {}^{i}H_j\,{}^{j}p$$

Dlaczego używamy współrzędnych jednorodnych?

Współrzędne jednorodne pozwalają zapisać translację i rotację w jednej macierzy oraz wykonywać ich składanie za pomocą zwykłego mnożenia macierzy. Jest to bardzo wygodne w robotyce, ponieważ każdy kolejny człon manipulatora, każdy kolejny układ lokalny lub każdy kolejny element narzędzia może być opisany przez własną macierz, a cały łańcuch kinematyczny otrzymujemy przez kolejne mnożenia.

1.3 Ćwiczenia – przestrzeń 2D

Ćwiczenie 1: Transformacja punktu na płaszczyźnie

Zaczniemy od najprostszego przypadku, czyli od transformacji pojedynczego punktu na płaszczyźnie. Celem tego ćwiczenia jest zrozumienie, że macierz przekształcenia jednorodnego jednocześnie obraca punkt oraz przesuwa go do nowego położenia.

Najpierw definiujemy funkcję tworzącą macierz przekształcenia w przestrzeni 2D.

```
H2[x_, y_, phi_] := {
 {Cos[phi], -Sin[phi], x},
 {Sin[phi],  Cos[phi], y},
 {0, 0, 1}
}
```

W powyższej funkcji: parametry `x` oraz `y` odpowiadają translacji, natomiast parametr `phi` odpowiada rotacji.

Rozważmy punkt zapisany w postaci jednorodnej oraz przykładową transformację:

```
pLocal = {1, 1, 1};

H = H2[2, 1, Pi/4];

pGlobal = H.pLocal
```

Wynik `pGlobal` jest nowym położeniem punktu po wykonaniu obrotu i przesunięcia.

Aby lepiej zrozumieć wynik, narysujmy punkt przed i po transformacji.

```
Graphics[
 {
  Blue, PointSize[Large], Point[pLocal[[1 ;; 2]]],
  Red, PointSize[Large], Point[pGlobal[[1 ;; 2]]]
 },
 Axes -> True,
 GridLines -> Automatic,
 PlotRange -> {{-1, 5}, {-1, 5}}
]
```

Na wykresie:

- niebieski punkt przedstawia położenie początkowe,
- czerwony punkt przedstawia położenie po transformacji.

W tym ćwiczeniu student powinien zauważyć, że transformacja nie działa jak samo przesunięcie ani jak sama rotacja, lecz jako ich połączenie.

Ćwiczenie 2: Interpretacja układu współrzędnych w 2D

W poprzednim ćwiczeniu transformowaliśmy pojedynczy punkt. Teraz przejdziemy o krok dalej i pokażemy, że macierz jednorodna opisuje tak naprawdę cały układ współrzędnych, a nie tylko ruch jednego obiektu.

Jeżeli mamy macierz ${}^{0}H_1$, to oznacza ona położenie i orientację układu 1 względem układu bazowego 0.

Zdefiniujmy funkcję rysującą osie układu współrzędnych w 2D.

```
axes2[H_] :=
 {
  Thick, Red, Arrow[{H[[1 ;; 2, 3]], H[[1 ;; 2, 3]] + H[[1 ;; 2, 1]]}],
  Thick, Green, Arrow[{H[[1 ;; 2, 3]], H[[1 ;; 2, 3]] + H[[1 ;; 2, 2]]}]
 }
```

Wektor kolumny pierwszej odpowiada kierunkowi osi X, wektor kolumny drugiej odpowiada kierunkowi osi Y, natomiast kolumna trzecia zawiera położenie początku układu.

Rozważmy przykładową transformację:

```
H01 = H2[2, 1, Pi/4];
```

Narysujmy układ bazowy oraz układ przekształcony.

```
Graphics[
 {
  axes2[IdentityMatrix[3]],
  axes2[H01]
 },
 Axes -> True,
```

```
 GridLines -> Automatic,
 PlotRange -> {{-1, 5}, {-1, 5}}
]
```

Na wykresie widać dwa układy współrzędnych:

- układ bazowy umieszczony w początku,
- układ 1, który został obrócony i przesunięty.

To ćwiczenie jest bardzo ważne, ponieważ w robotyce najczęściej nie transformujemy pojedynczych punktów, ale przechodzimy pomiędzy kolejnymi układami związanymi z członami robota.

Ćwiczenie 3: Składanie przekształceń w 2D

W praktyce bardzo rzadko występuje tylko jedno przekształcenie. Znacznie częściej mamy kilka kolejnych układów: na przykład układ bazowy, układ pierwszego członu oraz układ drugiego członu.

W takim przypadku korzystamy ze składania macierzy:

$$^{0}H_2 = {}^{0}H_1\,{}^{1}H_2$$

Zdefiniujmy dwa przekształcenia:

```
H01 = H2[1, 0, Pi/6];
H12 = H2[0, 2, Pi/4];

H02 = H01.H12;
```

Macierz `H01` opisuje układ 1 względem układu 0, a macierz `H12` opisuje układ 2 względem układu 1. Ich iloczyn daje położenie układu 2 względem układu 0.

Narysujmy wszystkie trzy układy:

```
Graphics[
 {
  axes2[IdentityMatrix[3]],
  axes2[H01],
  axes2[H02]
 },
 Axes -> True,
 GridLines -> Automatic,
 PlotRange -> {{-2, 5}, {-1, 5}}
]
```

Możemy także sprawdzić, jak położenie punktu zmienia się po zastosowaniu tych przekształceń.

```
p2 = {1, 0, 1};

p0 = H02.p2;
```

Teraz narysujmy punkt lokalny oraz jego obraz w układzie bazowym.

```
Graphics[
 {
  axes2[IdentityMatrix[3]],
  axes2[H01],
  axes2[H02],
  Blue, PointSize[Large], Point[H02[[1 ;; 2, 3]]],
  Red, PointSize[Large], Point[p0[[1 ;; 2]]]
```

```
 },
 Axes -> True,
 GridLines -> Automatic,
 PlotRange -> {{-2, 6}, {-1, 5}}
]
```

W tym ćwiczeniu student powinien zauważyć, że kolejne przekształcenia tworzą łańcuch, a wynik końcowy zależy od ich kolejności. Mnożenie macierzy transformacji nie jest przemienne, dlatego zmiana kolejności daje inny rezultat.

Ćwiczenie 4: Punkt lokalny i punkt globalny w 2D

W tym ćwiczeniu chcemy wyraźnie odróżnić punkt opisany w układzie lokalnym od tego samego punktu opisanego w układzie globalnym. Jest to bardzo ważne, ponieważ w robotyce często znamy pozycję punktu we współrzędnych narzędzia lub członu, a chcemy wyznaczyć jego pozycję w układzie bazowym.

Przyjmijmy, że punkt ma współrzędne lokalne:

```
p1 = {1, 0.5, 1};
```

oraz że układ 1 jest opisany względem układu 0 przez:

```
H01 = H2[2, 1, Pi/3];
```

Współrzędne globalne punktu obliczamy jako:

```
p0 = H01.p1
```

Aby pokazać, gdzie punkt znajduje się lokalnie i gdzie pojawia się w układzie globalnym, narysujmy układ bazowy, układ lokalny oraz punkt po transformacji.

```
Graphics[
 {
  axes2[IdentityMatrix[3]],
  axes2[H01],
  Blue, PointSize[Large], Point[H01[[1 ;; 2, 3]]],
  Red, PointSize[Large], Point[p0[[1 ;; 2]]]
 },
 Axes -> True,
 GridLines -> Automatic,
 PlotRange -> {{-1, 5}, {-1, 5}}
]
```

Tutaj czerwony punkt nie jest już opisany w lokalnym układzie 1, lecz w układzie bazowym 0. To dokładnie ten typ operacji, który wykonuje się podczas wyznaczania położenia końcówki robota.

1.4 Ćwiczenia – przestrzeń 3D

Ćwiczenie 5: Podstawowe macierze przekształceń jednorodnych w 3D

Po zrozumieniu przypadku płaskiego możemy przejść do przestrzeni trójwymiarowej. W przestrzeni 3D również korzystamy z macierzy przekształceń jednorodnych, jednak ich rozmiar wynosi teraz 4×4.

W praktyce wygodnie jest budować bardziej złożone przekształcenia z prostych, elementarnych ruchów. Dlatego zamiast definiować od razu jedną ogólną macierz, wprowadzimy sześć podstawowych macierzy przekształceń jednorodnych, z których będziemy dalej korzystać.

Będą to macierze odpowiadające:

- translacji wzdłuż osi X,
- translacji wzdłuż osi Y,
- translacji wzdłuż osi Z,
- rotacji wokół osi X,
- rotacji wokół osi Y,
- rotacji wokół osi Z.

Takie podejście jest bardziej przejrzyste, ponieważ student od razu widzi, jaki konkretny ruch opisuje dana macierz. Później z tych prostych przekształceń będziemy budować przekształcenia bardziej złożone.

Zdefiniujmy sześć podstawowych macierzy.

```
Hx[d_] := {
 {1, 0, 0, d},
 {0, 1, 0, 0},
 {0, 0, 1, 0},
 {0, 0, 0, 1}
};

Hy[d_] := {
 {1, 0, 0, 0},
 {0, 1, 0, d},
 {0, 0, 1, 0},
 {0, 0, 0, 1}
};

Hz[d_] := {
 {1, 0, 0, 0},
 {0, 1, 0, 0},
 {0, 0, 1, d},
 {0, 0, 0, 1}
};

Hrx[a_] := {
 {1, 0, 0, 0},
 {0, Cos[a], -Sin[a], 0},
 {0, Sin[a], Cos[a], 0},
 {0, 0, 0, 1}
};

Hry[a_] := {
 {Cos[a], 0, Sin[a], 0},
 {0, 1, 0, 0},
 {-Sin[a], 0, Cos[a], 0},
 {0, 0, 0, 1}
};

Hrz[a_] := {
 {Cos[a], -Sin[a], 0, 0},
 {Sin[a], Cos[a], 0, 0},
 {0, 0, 1, 0},
 {0, 0, 0, 1}
};
```

Rozważmy teraz punkt zapisany w postaci jednorodnej:

```
pLocal = {1, 0, 0, 1};
```

Wykonajmy przekształcenie złożone, w którym punkt zostanie najpierw obrócony wokół osi Z, a następnie przesunięty o 1 wzdłuż osi X, o 2 wzdłuż osi Y oraz o 1 wzdłuż osi Z.

Ponieważ korzystamy z wektorów kolumnowych, przekształcenia wykonywane są od prawej do lewej. Oznacza to, że macierz stojąca najbliżej wektora działa jako pierwsza.

```
H01 = Hx[1].Hy[2].Hz[1].Hrz[Pi/4];

pGlobal = H01.pLocal
```

Wynik `pGlobal` jest położeniem punktu w układzie bazowym po wykonaniu wszystkich przekształceń.

Aby łatwiej zinterpretować ten wynik, narysujmy punkt przed i po transformacji.

```
Graphics3D[
 {
  Blue, PointSize[Large], Point[pLocal[[1 ;; 3]]],
  Red, PointSize[Large], Point[pGlobal[[1 ;; 3]]]
 },
 Axes -> True,
 Boxed -> True,
 PlotRange -> {{-2, 3}, {-2, 4}, {-1, 3}}
]
```

Na wykresie:

- niebieski punkt przedstawia położenie początkowe,
- czerwony punkt przedstawia położenie po transformacji.

W tym ćwiczeniu najważniejsze jest zrozumienie, że nawet dość złożone przekształcenie w przestrzeni można zbudować z prostych, elementarnych ruchów.

Ćwiczenie 6: Układy współrzędnych w 3D

W przestrzeni 3D szczególnie istotna jest nie tylko pozycja, ale również orientacja układu współrzędnych. Macierz przekształcenia jednorodnego opisuje cały układ: położenie jego początku oraz kierunki wszystkich trzech osi.

Aby to zobaczyć, zdefiniujmy funkcję rysującą układ współrzędnych w 3D.

```
axes3[H_] :=
 {
  Thick, Red, Arrow[{H[[1 ;; 3, 4]], H[[1 ;; 3, 4]] + H[[1 ;; 3, 1]]}],
  Thick, Green, Arrow[{H[[1 ;; 3, 4]], H[[1 ;; 3, 4]] + H[[1 ;; 3, 2]]}],
  Thick, Blue, Arrow[{H[[1 ;; 3, 4]], H[[1 ;; 3, 4]] + H[[1 ;; 3, 3]]}]
 }
```

Kolumny pierwsza, druga i trzecia opisują kierunki osi układu lokalnego, natomiast kolumna czwarta określa położenie początku tego układu.

Zbudujmy teraz przekształcenie jako złożenie elementarnych translacji i rotacji. Nie korzystamy tutaj z żadnej ogólnej funkcji, lecz wyłącznie z wcześniej zdefiniowanych sześciu macierzy.

```
H01 = Hx[1].Hy[1].Hz[1].Hrz[Pi/4].Hry[Pi/6];
```

W tej transformacji: układ zostaje najpierw obrócony, a następnie przesunięty. Ponownie należy pamiętać, że przy wektorach kolumnowych rzeczywista kolejność wykonywania przekształceń jest od prawej do lewej.

Narysujmy układ bazowy oraz układ przekształcony.

```
Graphics3D[
 {
  axes3[IdentityMatrix[4]],
  axes3[H01]
 },
 Axes -> True,
 Boxed -> True,
 PlotRange -> {{-2, 3}, {-2, 3}, {-2, 3}}
]
```

Na wykresie widoczne są dwa układy współrzędnych:

- układ bazowy,
- układ przekształcony.

Ćwiczenie to pokazuje, że orientację układu w przestrzeni 3D można budować przez składanie prostych rotacji wokół osi, a położenie początku układu przez składanie prostych translacji.

Ćwiczenie 7: Składanie przekształceń w 3D

Podobnie jak w przestrzeni 2D, również w przestrzeni 3D możemy składać kolejne transformacje. Jest to szczególnie ważne w robotyce, ponieważ każdy kolejny człon manipulatora może być opisany przez osobną macierz przekształcenia.

Rozważmy dwa kolejne przekształcenia zbudowane wyłącznie z elementarnych translacji i rotacji.

```
H01 = Hx[1].Hrz[Pi/4];
H12 = Hy[1].Hz[1].Hrx[Pi/6];

H02 = H01.H12;
```

Macierz `H01` opisuje układ 1 względem układu 0, macierz `H12` opisuje układ 2 względem układu 1, a ich iloczyn `H02` opisuje układ 2 względem układu 0.

Narysujmy wszystkie trzy układy współrzędnych.

```
Graphics3D[
 {
  axes3[IdentityMatrix[4]],
  axes3[H01],
  axes3[H02]
 },
 Axes -> True,
 Boxed -> True,
 PlotRange -> {{-2, 3}, {-2, 3}, {-2, 3}}
]
```

Aby jeszcze lepiej zrozumieć sens tego złożenia, rozważmy punkt lokalny związany z układem 2:

```
p2 = {0.5, 0, 0, 1};

p0 = H02.p2
```

Punkt `p2` jest opisany we współrzędnych układu 2, natomiast wynik `p0` przedstawia ten sam punkt w układzie bazowym 0.

Zaznaczmy go na wykresie.

```
Graphics3D[
 {
  axes3[IdentityMatrix[4]],
  axes3[H01],
  axes3[H02],
  Black, PointSize[Large], Point[p0[[1 ;; 3]]]
 },
 Axes -> True,
 Boxed -> True,
 PlotRange -> {{-2, 3}, {-2, 3}, {-2, 3}}
]
```

W tym ćwiczeniu student powinien zwrócić uwagę na to, że całą logikę budowania łańcucha przekształceń w 3D można oprzeć wyłącznie na prostych macierzach elementarnych. Każda z nich opisuje pojedynczy, łatwy do interpretacji ruch, a ich iloczyn daje przekształcenie końcowe.

1.5 Zadania do samodzielnego rozwiązania

Poniższe zadania mają na celu utrwalenie sposobu opisu położenia i orientacji z wykorzystaniem macierzy przekształceń jednorodnych. W każdym zadaniu należy wykonać obliczenia, a następnie przygotować odpowiednią wizualizację.

Zadanie 1 – transformacja punktu

Rozważ punkt:

$$p = (1, 2)$$

oraz transformację:

$$H = \begin{bmatrix} \cos 45^\circ & -\sin 45^\circ & 2 \\ \sin 45^\circ & \cos 45^\circ & 1 \\ 0 & 0 & 1 \end{bmatrix}$$

Oblicz nową pozycję punktu oraz przedstaw wynik graficznie.

Zadanie 2 – składanie transformacji

Zdefiniuj dwie transformacje:

$$^0H_1 : \quad x = 1,\ y = 0,\ \varphi = 30^\circ$$

$$^1H_2 : \quad x = 0,\ y = 2,\ \varphi = 60^\circ$$

Następnie:

- oblicz macierz $^0H_2$,
- zastosuj otrzymane przekształcenie do punktu $(1, 0)$,
- porównaj wynik z wykonaniem transformacji krok po kroku,
- przedstaw wszystkie etapy na wykresie.

Zadanie 3 – wizualizacja układów współrzędnych

Narysuj dwa układy współrzędnych na płaszczyźnie:

- układ bazowy,
- układ przekształcony o translację $(2, 1)$ i rotację $45°$.

Następnie opisz, jak zmieniło się położenie początku układu oraz jak zmieniła się orientacja jego osi.

Zadanie 4 – robot 2DoF

Rozważ robota planarnego o dwóch stopniach swobody.

Rysunek 1: Robot 2DoF – schemat (należy uzupełnić)

Załóż:

$$l_1 = 1, \quad l_2 = 1$$

oraz kąty:

$$\theta_1, \quad \theta_2$$

Następnie:

- zdefiniuj macierz ${}^0H_1$,
- zdefiniuj macierz ${}^1H_2$,
- oblicz macierz końcową ${}^0H_2$,
- wyznacz pozycję efektora,
- wyznacz orientację efektora,
- wykonaj wizualizację dla wybranych wartości θ_1 i θ_2.

Zadanie 5 – animacja ruchu robota

Dla robota z poprzedniego zadania:

- utwórz animację zmiany θ_1 i θ_2,
- narysuj trajektorię końcówki robota,
- spróbuj określić obszar roboczy manipulatora.

Zadanie 6 – transformacja punktu w przestrzeni 3D

Rozważ punkt:

$$p = (1, 1, 1)$$

Wykonaj jego transformację w przestrzeni 3D dla macierzy składającej się z:

- rotacji wokół osi Z o $45°$,
- translacji o wektor $(1, 2, 1)$.

Następnie:

- oblicz nowe współrzędne punktu,
- narysuj punkt przed transformacją,
- narysuj punkt po transformacji,
- zinterpretuj geometrycznie otrzymany wynik.

Zadanie 7 – wizualizacja układów współrzędnych w 3D

Narysuj dwa układy współrzędnych w przestrzeni:

- układ bazowy,
- układ przekształcony przez wybraną translację oraz złożenie dwóch rotacji.

Spróbuj odpowiedzieć na pytania:

- jak zmienia się orientacja osi układu,
- czy kolejność wykonywania rotacji ma znaczenie,
- które elementy macierzy odpowiadają za położenie,
- które elementy macierzy odpowiadają za orientację.